# Resolução dos exercícios propostos referentes à aula 4

```julia
# Carregando pacotes necessários
begin
    using DelimitedFiles
    using Plots
end
```

## Exercício 1

No experimento realizado foi admitido que o arrasto aerodinâmico (força de resistência ao avanço da bola) era desprezível. Sendo assim, com a gravidade constante, é de se esperar que os pontos experimentais se ajustem a uma parábola:

$$S(t) = C_0 + C_1 t + C_2 t^2$$

Da cinemática, as constantes correspondem ao espaço inicial, à velocidade inicial e à aceleração da gravidade do planeta em questão:

$$S(t) = S_0 + V_0 t + \frac{gt^2}{2}$$

Como a bola parte do chão, podemos considerar o espaço inicial nulo, com o vetor velocidade e o vetor aceleração positivos para cima, ou seja:

$$S(t) = V_0 t - \frac{gt^2}{2} = C_1 t + C_2 t^2$$

Para encontrar o melhor ajuste, conforme visto em sala de aula, podemos resolver o sistema

$$A^T A \hat{x} = A^T b$$

sendo $A$ a matrix com a primeira coluna igual aos valores de tempo medidos no experimento e a segunda coluna igula aos valores de tempo ao quadrado, $\hat{x}$ o vetor com os coeficientes da equação de segundo grau e $b$ os espaços medidos no experimento, conforme indicado abaixo.

```
(10×2 Matrix{Float64}:, 1×2 Matrix{AbstractString}:)
  0.5   5.68             "t(s)"  "S(m)"
  1.0   7.68
  1.5  10.91
  2.0  12.29
  2.5  13.71
  3.0  13.54
  3.5  11.87
  4.0  10.47
  4.5   6.76
  5.0   3.86
```

```julia
# Carregando dados experimentais
d1,_ = readdlm("dados_p4_1.txt"; header=true)
```

```
A = 10×2 Matrix{Float64}:
     0.5   0.25
     1.0   1.0
     1.5   2.25
     2.0   4.0
     2.5   6.25
     3.0   9.0
     3.5  12.25
     4.0  16.0
     4.5  20.25
     5.0  25.0
```

```julia
A = [d1[:,1] d1[:,1] .^ 2]
```

```
C =   Float64[
          1:  10.038
          2:  -1.86883
     ]
```

```julia
# Resolvendo o sistema linear
C = A'*A \ A'*d1[:,2]
```

Note que os coeficientes encontrados são:

$$C_1 = 10,038 \rightarrow V_0$$

$$C_2 = -1,8688 \rightarrow -g/2$$

Logo, a velocidade inicial da bola é de aproximadamente 10 m/s para cima e a gravidade do planeta em questão vale aproximadamente 3,74 m/s$^2$.

Sendo assim, é provável que o experimento tenha sido realizado em **Marte**!

Vamos plotar os resultados para verificar se o ajuste obtido é razoável.




```
begin
    scatter(d1[:,1], d1[:,2], markersize=6, label="medidas")
    xlabel!("t(s)")
    ylabel!("S(m)")
    s(t) = C[1]*t + C[2]*t^2
    plot!(0.0:0.1:5.3, s.(0.0:0.1:5.3), label="ajuste", linewidth=2)
end
```

# Exercício 2

O segundo exercício trata do ajuste de curva para a região da camada limite atmosférica (no caso modelada em um túnel de vento) em que é válida a lei logarítmica indicada abaixo.

$$U(z) = \frac{u^*}{\kappa} ln\left(\frac{z}{z_0}\right)$$

sendo $U$ os valores de velocidade medidos com um tubo de Pitot para cada altura $z$, $u^*$ a velocidade de atrito, $\kappa$ a constante de von Kármán e $z_0$ o comprimento de rugosidade. Deseja-se estimar a velocidade de atrito e principalmente o comprimento de rugosidade, que será necessário para verificar se está sendo simulada a categoria de terreno adequada.

Manipulando a equação da lei log, lembrando das propriedades de funções logarítmicas, chegamos à seguinte equação:

$$U = \frac{u^*}{\kappa} ln(z) - \frac{u^*}{\kappa} ln(z_0)$$

O segundo termo à direita da equação é um valor constante. Se substituirmos a variável $ln(z)$ por $x$ (ou plotarmos em um gráfico monolog), acabamos linearizando a equação, ou seja,

$$U = Cx + D$$

sendo $C = u^*/\kappa$, $x = ln(z)$ e $D = -Cln(z_0)$.

Vamos carregar e plotar os pontos experimentais.

```
(39×2 Matrix{Float64}:, 1×2 Matrix{AbstractString}:)
   10.0   6.73           "z(mm)"  "U(m/s)"
   15.0   7.17
   20.0   7.17
   25.0   7.21
   30.0   7.56
   35.0   7.71
   40.0   7.78
    ⋮
  400.0  14.16
  420.0  14.38
  460.0  14.83
  500.0  14.96
  540.0  15.46
  580.0  15.76
```

```
# Carregando os dados
d2,_ = readdlm("dados_p4_2.txt", header=true)
```

```
lnz =
[-4.60517, -4.19971, -3.91202, -3.68888, -3.50656, -3.35241, -3.21888, -3.10109, -2.9957
```

```
# Convertendo tudo para o SI e tirando o logaritmo
lnz = log.(d2[:,1] ./ 1000)
```




```
scatter(d2[:,2], lnz, markersize=5, xlabel="U", ylabel="ln(z)", label="")
```

É possível notar que os primeiros pontos experimentais não obedecem a lei log. A partir do 12º ponto, aproximadamente, podemos ver que a relação se aproxima a uma reta, como experado, indicando que a partir desse ponto devemos ter um ajuste bom pelo modelo log.

Vamos então considerar a partir do 12º ponto experimental.

```
[9.27, 9.56, 9.88, 10.12, 10.2, 10.36, 10.54, 10.99, 11.31, 11.4, 11.57, 11.94, 11.77, 12
```

```
begin
    x = lnz[12:end]
    U = d2[12:end,2]
end
```

Para resolver este problema vamos utilizar o pacote LsqFit.jl. O mesmo procedimento adotado no exercício anterior e em sala de aula poderia ter sido utilizado.

```
using LsqFit
```

```
LsqFitResult([3.29062, 17.2916], [-0.289571, -0.191991, -0.16529, -0.0916599, 0.114662,
```

```
# Criando modelo
begin
    model(z,p) = p[1]*z .+ p[2] # modelo
    p0 = [2.0, 10.0] # parâmetros iniciais (chute)
    fit = curve_fit(model, x, U, p0) # ajuste
end
```

```
Float64[
    1:  3.29062
    2:  17.2916
]
```

```
fit.param
```

As constantes encontradas estão indicadas acima. Com isso podemos calcular a velocidade de atrito (sabendo que a constante de von Kármán vale 0,4) e o comprimento de rugosidade (relembrar regrinhas de logaritmos).

$$u^* = 0,4C$$

$$z_0 = exp\left(-\frac{D}{C}\right)$$

```
u_atrito = 1.3162486113732523
```

```
u_atrito = 0.4*fit.param[1]
```

```
z₀ = 0.005222250235330567
```

```
z₀ = exp(-fit.param[2]/fit.param[1])
```

Ou seja, a **velocidade de atrito vale aproximadamente 1,32 m/s** e o **comprimento de rugosidade aproximadamente 5,22 mm**. O ideal é sempre trabalhar no SI, como fizemos no início do problema. De qualquer forma, como neste caso as alturas e o comprimento de rugosidade aparecem divididos, o único ponto importante é que estejam na mesma unidade. A velocidade de atrito, por sua vez, terá a mesma unidade das velocidades medidas.

Vamos plotar os resultados em gráficos para verificar o ajuste.




```
begin
    scatter(d2[1:11,2], lnz[1:11], markersize=5, xlabel="U", ylabel="ln(z)",
    label="")
    scatter!(U, x, markersize=5, label="")
    Plots.abline!(1/fit.param[1], -fit.param[2]/fit.param[1], line=:dash,
    linewidth=3, label="")
end
```




```
begin
    cl(z) = u_atrito/0.4 * log(z/z₀)
    scatter(d2[:,2], d2[:,1] ./1000, markersize=5, xlabel="U(m/s)", ylabel="z(mm)",
    label="")
    plot!(cl.(0.03:0.01:0.6), 0.03:0.01:0.6, linewidth=2, label="")
end
```

Apenas como exemplo, se o modelo ensaiado estiver na escala 1:200, o $z_0$ na escala real vale aproximadamente 1,0 m (0,005 x 200), o que corresponde à categoria de terrena IV de acordo com a norma brasileira NBR 6123.